

MK1 "Calcul formel" Maple

TP3 : Résolution d'équations, dérivation, intégration

Version de démonstration de Maple :

Vous pouvez installer une version de démonstration de Maple (gratuite, mais limitée) sur votre ordinateur personnel, quelque soit son système d'exploitation (Windows, Linux, MacOS,...) : il suffit de la télécharger sur le ftp de Maplesoft, la société qui édite Maple :

<ftp://ftp.maplesoft.com/pub/maple/demo>

La version de Maple proposée est V Release 4. En TP, vous utilisez la version 9. Il existe de (petites) différences entre les deux.

Commandes pour les caractères spéciaux

Voici les commandes pour obtenir les caractères qui ne sont pas présents sur les claviers Mac.

pour ~ : Alt n
pour { : Alt (pour [: Alt Shift (pour } : Alt)
pour } : Alt) pour] : Alt Shift)

Et surtout, n'oubliez pas de vous (et de me) poser des questions !

1. Résoudre des équations

1.1 Résoudre une équation à une inconnue

Pour résoudre une équation à une inconnue, on utilise la commande *solve* dont la syntaxe est *solve(equation, variable)*.

L'équation est définie par une égalité =, à bien différencier des affectations, qui se font par :=.

Commençons par les **équations polynomiales**. Maple résout complètement sur C les équations polynomiales de degré <=3 et quelques équations de degré supérieur.

```
> restart;
solve(x=1+1/x, x);
```

$$\frac{1}{2} + \frac{\sqrt{5}}{2} \cdot \frac{1}{2} - \frac{\sqrt{5}}{2}$$

Ici, Maple donne les solutions exactes et explicites de l'équation. Le type de résultat renvoyé par Maple est une *séquence*. On peut demander les valeurs approchées des solutions par :

```
> evalf(%);
```

1.618033988, -0.6180339880

Malheureusement, en degré supérieur ou égal à 4, cela ne se passe pas toujours aussi bien.

Essayons de résoudre l'équation : $x^7 - x^6 + x^2 - 1 = 0$.

```
> S:=solve(x^7-x^6+x^2-1=0, x);
```

```
S := 1, RootOf(_Z^6 + _Z + 1, index=1), RootOf(_Z^6 + _Z + 1, index=2),
      RootOf(_Z^6 + _Z + 1, index=3), RootOf(_Z^6 + _Z + 1, index=4),
      RootOf(_Z^6 + _Z + 1, index=5), RootOf(_Z^6 + _Z + 1, index=6)
```

Parfois, lorsque Maple ne peut pas trouver de solution explicite, ou lorsque celle-ci est trop compliquée pour être utilisable, les solutions sont exprimées à l'aide de la fonction *RootOf* ("racine de"), qui "représente" toutes les racines à la fois et aucune en particulier. Factorisons le polynôme de départ.

```
> factor(x^7-x^6+x^2-1);
```

$$(x-1)(x^6+x+1)$$

Une racine évidente est 1, les autres racines sont celles de x^6+x+1 (au nombre de 6), que Maple refuse d'expliciter. Cependant, ce n'est pas complètement un échec car :

* certaines fonctions de Maple sont capables de travailler avec *RootOf*, et on peut continuer alors à "faire des calculs" :

```
> alias(alpha=RootOf(x^6+x+1));
```

α

```
> alpha^8;
simplify(alpha^8);
```

α^8

$$-\alpha^2(1+\alpha)$$

* si l'équation ne comporte aucun paramètre, on peut obtenir des valeurs numériques approchées des racines :

```
> evalf(S);
```

```
1., 0.9454023333 + 0.6118366938 I, -0.1547351445 + 1.038380754 I,
-0.7906671888 + 0.3005069203 I, -0.7906671888 - 0.3005069203 I,
-0.1547351445 - 1.038380754 I, 0.9454023333 - 0.6118366938 I
```

La fonction *allvalues* permet parfois de "déterminer" toutes les racines désignées par le *RootOf* : quand c'est possible, Maple donne des expressions exactes.

```
> T:=solve(-x**4+5*x**3+x**2-1, x);
```

```
T := RootOf(_Z^4 - 5 _Z^3 - _Z^2 + 1, index=1), RootOf(_Z^4 - 5 _Z^3 - _Z^2 + 1, index=2),
      RootOf(_Z^4 - 5 _Z^3 - _Z^2 + 1, index=3), RootOf(_Z^4 - 5 _Z^3 - _Z^2 + 1, index=4)
```

```
> allvalues([T]):
allvalues([T])[1];
```

$$\frac{5}{4} + \frac{\sqrt{3} \sqrt{\frac{83(2980+12\sqrt{60693})^{(1/3)} + 2(2980+12\sqrt{60693})^{(2/3)} + 104}{(2980+12\sqrt{60693})^{(1/3)}}}}{12} - \left(\left(498 \right. \right.$$

$$\left. \left. \sqrt{\frac{(2980+12\sqrt{60693})^{(1/3)} \sqrt{83(2980+12\sqrt{60693})^{(1/3)} + 2(2980+12\sqrt{60693})^{(2/3)} + 104}}{(2980+12\sqrt{60693})^{(1/3)}}} - 6 \right) \right)$$

$$\sqrt{\frac{83(2980+12\sqrt{60693})^{(1/3)}+2(2980+12\sqrt{60693})^{(2/3)}+104}{(2980+12\sqrt{60693})^{(1/3)}}}$$

$$-312\sqrt{\frac{83(2980+12\sqrt{60693})^{(1/3)}+2(2980+12\sqrt{60693})^{(2/3)}+104}{(2980+12\sqrt{60693})^{(1/3)}}}$$

$$+2610\sqrt{3(2980+12\sqrt{60693})^{(1/3)}} \Big/ \left((2980+12\sqrt{60693})^{(1/3)} \right)$$

$$\sqrt{\frac{83(2980+12\sqrt{60693})^{(1/3)}+2(2980+12\sqrt{60693})^{(2/3)}+104}{(2980+12\sqrt{60693})^{(1/3)}}} \Big)^{(1/2)} / 12$$

Maple peut aussi résoudre quelques **équations non polynomiales** simples.

```
> solve (exp (x)=1+x, x);
```

0

Pour les équations plus compliquées, Maple peut trouver une partie des solutions, aucune des solutions, ou même des solutions qui sont fausses ! Voir la feuille d'exercices.

1.2 Résoudre un système d'équations

Pour résoudre un système d'équations, on utilise la commande `solve` avec la syntaxe :

```
solve({equations},{variables})
```

Maple retourne les solutions sous la forme {variable1=expression1, variable2=expression2,...}.

Exemple : le système {x+y+z=4, x+y-z=1, x-y-z=-3}.

```
> solve ({x+y+z=4, x+y-z=1, x-y-z=-3}, {x, y, z});
```

$$\left\{ x = \frac{1}{2}, z = \frac{3}{2}, y = 2 \right\}$$

Ici, Maple trouve une solution unique.

Dans certains cas, on a une famille de solutions, qui s'expriment en fonction de certaines des variables.

Exemple : le système {x-3y=2, 3x-9y=6}.

```
> solve ({x-3*y=2, 3*x-9*y=6});
```

$$\{x = 3y + 2, y = y\}$$

Maple a choisi de garder l'inconnue y comme paramètre, et a pu exprimer toutes les solutions en fonction de ce paramètre.

1.3 Résolution approchée

Lorsqu'on ne peut résoudre de manière exacte une équation (sans paramètre), on peut essayer la fonction `fsolve` pour obtenir des solutions *approchées*.

```
> fsolve (tan (sin (x))=1, x);
```

0.9033391108

On peut spécifier un intervalle sur lequel on cherche les solutions :

```
> fsolve (x^5-7*x^4+3*x^2+15, x, 0..10);
```

1.386962979, 6.931051639

Par défaut, Maple cherche des solutions approchées réelles. On peut aussi lui demander des solutions approchées complexes :

```
> fsolve (x^5-7*x^4+3*x^2+15, x, complex);
```

-1.242272278, -0.03787117016 - 1.120100886 I, -0.03787117016 + 1.120100886 I,

1.386962979, 6.931051639

2. Dérivation et intégration

2.1 Dériver une expression

On dérive une *expression* à l'aide de la commande `diff`, par la syntaxe `diff(expression_en_variable,variable)`.

```
> diff (1/(1+x^2), x);
```

$$-\frac{2x}{(1+x^2)^2}$$

Maple retourne alors une *expression* en la variable.

2.2 Dériver une fonction

On dérive une *fonction* (préalablement définie à l'aide de `->` ou de `unapply`) par l'opérateur `D`.

```
> restart; f:=x->cos(x)+5*x^2-3;
```

$$f := x \rightarrow \cos(x) + 5x^2 - 3$$

```
> g:=D(f);
```

$$g := x \rightarrow -\sin(x) + 10x$$

Maple retourne alors une *fonction*.

```
> g(Pi);
```

10 π

2.3 Calculer une primitive

On utilise la commande `int`, dont la syntaxe est `int(expression_en_variable,variable)`. Le résultat est sans constante d'intégration. C'est une expression en la variable.

```
> int (tan (x), x);
```

$$-\ln(\cos(x))$$

Parfois, Maple ne sait pas calculer exactement une primitive.

```
> int (exp (x) *cos (x) ^n, x);
```

$$\int e^x \cos(x)^n dx$$

Vérifier qu'ici, si l'on donne une valeur précise à n, Maple sait mener le calcul.


Parfois, Maple exprime les primitives en fonction de fonctions dites spéciales :

```
> int (sin (x) /x, x);
```

Si(x)

C'est la fonction sinus intégral (pour en savoir plus, consulter l'aide sur Si). Maple la connaît, il sait par exemple tracer le graphe :

```
> plot (Si);
```



2.4 Calculer une intégrale

Pour calculer l'intégrale de a à b de f , on utilise la commande `int` avec la syntaxe

`int(expression_en_variable, variable=a..b).`

```
> restart; int (exp(x)*cos(x)^3, x=0..Pi/2);
```

$$\frac{3}{10} e^{\left(\frac{\pi}{2}\right)} - \frac{2}{5}$$

[Les bornes ne sont pas nécessairement des constantes. Elles peuvent être des indéterminées :

```
> int (x^2*cos(x), x=0..N*Pi);
```

$$N^2 \pi^2 \sin(N \pi) - 2 \sin(N \pi) + 2 N \pi \cos(N \pi)$$

[Elles peuvent être aussi +l'infini (noté `infinity`) ou -l'infini (noté `-infinity`).

```
> int (exp(-x^2)*cos(x), x=0..infinity);
```

$$\frac{1}{2} \sqrt{\pi} e^{(-1/4)}$$

[(mais vous n'aurez pas de telles intégrales dans votre cours de mathématiques du premier semestre)

FEUILLE D'EXERCICES N°3

EXERCICE 1. Résoudre l'équation $x^2 + 2bx + c = 0$ d'inconnue x et de paramètres b et c . Donner la liste des solutions. En utilisant la commande *sols*, trouver les solutions pour $b = 1$ et $c = 2$. Que constatez-vous ? Que remarquez-vous sur les solutions générales de l'équation ?

EXERCICE 2. Résoudre l'équation $x^3 - 5x^2 - 2 = 0$. Donner une valeur approchée de la première solution donnée par Maple.

EXERCICE 3. Cet exercice a pour but de vous faire sentir les limites de Maple pour la résolution d'équations non polynomiales. Résolvez chaque équation et réfléchissez sur les résultats donnés par Maple.

- 1) Equation $\tan(x) = \sqrt{3}$.
- 2) Equation $\cos(x) = x$.
- 3) Equation $\sqrt{x^2} = x$.

EXERCICE 4. Chercher l'ensemble des nombres complexes z tels que $|z| = |z - 1|$.

EXERCICE 5.

- 1) Calculer les dérivées des fonctions suivantes : $x \mapsto \tan x$ et $x \mapsto \ln(x^3 + ax^2 + 1)$, où a est un paramètre réel.
- 2) Soient $f : x \mapsto 1 + x^5$ et $g : x \mapsto \ln(x)$. Calculer la dérivée de $g \circ f$.
- 3) Soit f comme précédemment. Construire la séquence des $f'(i)$, pour i de 1 à 20.

EXERCICE 6. Calculer les primitives des fonctions suivantes : $x \mapsto \frac{\sin(x) \tan(x) + \cos(x)}{\sin(x) - \cos(x)^2}$ et $x \mapsto \ln(x^2 + 1)$.

EXERCICE 7. Calculer $\int_0^1 e^{-t^2} dt$. Donner une valeur approchée de l'intégrale.